*===============================================================================
* Do-file creates "Table A1: Administrative Dataset: Violence and M-Paisa Use, All Users"
*===============================================================================
clear 
set more off

cd "~/Dropbox/VFD/REStat_Native_Files/Stata_Files"
local output "~/Dropbox/VFD/REStat_Native_Files/Stata_Files/Output"



use "monthpanel_analysis.dta", clear
gen trimmedobs=0
sum balance, d
replace trimmedobs =1 if (balance <=r(p99) & balance>=r(p1))
replace trimmedobs =. if balance ==.
bys identityid: egen trimmed_extusers = min(trimmedobs)
keep if trimmed_extusers==1


* Table A1, Panel A: Main violence
	foreach y in balance numtxns numwithdrawal numdeposit numtransferdebit { 
		reghdfe `y' d_violence_10km, absorb(ym identityid c.ym#district) cluster(district)
		est store `y'
		estadd ysumm, mean prefix(in_sample_)
		unique identityid if e(sample)
		estadd local nID = r(unique)
		estadd local samplestr "Salary Users"
		estadd local monthFE "YES"
		estadd local distFE "YES"
		estadd local individFE "YES"
	}
	estout balance numtxns numwithdrawal numdeposit numtransferdebit ///
	using "`output'/table_a1a.tex", cells(b(fmt(%9.3f) star) se(par fmt(%9.3f))) ///
	starlevels(* .1 ** .05 *** .01) style(tex) keep(d_violence_10km) mlabels(, dep) collabels(,none) ///
	stats(samplestr in_sample_mean nID N_clust N r2 monthFE individFE distFE, fmt(0 3 0 0 0 3 0 0 0) ///
	label("Sample" "Mean Dep Var" "\# Individuals" "\# Clusters" "\# Observations" "R-Squared" "Month-Year FE" "Individual FE" "District Trends")) ///
	label title("Administrative Dataset: Violence and M-Paisa Use") replace

* Table A1, Panel B: By violence categories
	foreach y in balance numtxns numwithdrawal numdeposit numtransferdebit { 
		reghdfe `y' d_df_10km d_idf_10km d_ied_10km, absorb(ym identityid c.ym#district) cluster(district)
		est store `y'
		estadd ysumm, mean prefix(in_sample_)
		unique identityid if e(sample)
		estadd local nID = r(unique)
		estadd local samplestr "All Users"
		estadd local monthFE "YES"
		estadd local distFE "YES"
		estadd local individFE "YES"
	}
	estout balance numtxns numwithdrawal numdeposit numtransferdebit ///
	using "`output'/table_a1b.tex", cells(b(fmt(%9.3f) star) se(par fmt(%9.3f))) ///
	starlevels(* .1 ** .05 *** .01) style(tex) keep(d_df_10km d_ied_10km d_idf_10km) mlabels(, dep) collabels(,none) ///
	stats(samplestr in_sample_mean nID N_clust N r2 monthFE individFE distFE, fmt(0 3 0 0 0 3 0 0 0) ///
	label("Sample" "Mean Dep Var" "\# Individuals" "\# Clusters" "\# Observations" "R-Squared" "Month-Year FE" "Individual FE" "District Trends")) ///
	label title("Administrative Dataset: Violence and M-Paisa Use") replace

	
	
